Beatty Sequences for a Quadratic Irrational: Decidability and Applications

Luke Schaeffer

Joint Center for Quantum Information and Computer Science 3100 Atlantic Building University of Maryland College Park, MD 20742-2420 USA

lrschaef@umd.edu

Jeffrey Shallit and Stefan Zorcic School of Computer Science University of Waterloo Waterloo, ON N2L 3G1 Canada shallit@uwaterloo.ca szorcic@uwaterloo.ca

March 26, 2024

Abstract

Let α and β belong to the same quadratic field. We show that the inhomogeneous Beatty sequence $(\lfloor n\alpha + \beta \rfloor)_{n\geq 1}$ is synchronized, in the sense that there is a finite automaton that takes as input the Ostrowski representations of n and y in parallel, and accepts if and only if $y = \lfloor n\alpha + \beta \rfloor$. Since it is already known that the addition relation is computable for Ostrowski representations based on a quadratic number, a consequence is a new and rather simple proof that the first-order logical theory of these sequences with addition is decidable. The decision procedure is easily implemented in the free software Walnut.

As an application, we show that for each $r \ge 1$ it is decidable whether the set $\{\lfloor n\alpha + \beta \rfloor : n \ge 1\}$ forms an additive basis (or asymptotic additive basis) of order r. Using our techniques, we also solve some open problems of Reble and Kimberling, and give an explicit characterization of a sequence of Hildebrand et al.

1 Introduction

Let α be an irrational number. The sequence $(\lfloor n\alpha \rfloor)_{n\geq 1}$ is known as a *Beatty sequence* [6] and has been very widely studied.¹ The two most famous Beatty sequences correspond to $\alpha = (1 + \sqrt{5})/2$ and $\alpha = (3 + \sqrt{5})/2$; they are also known as the *lower Wythoff* and *upper Wythoff* sequences, respectively, and they form sequences <u>A000201</u> and <u>A001950</u> in the *On-Line Encyclopedia of Integer Sequences* (OEIS) [46].

In an *inhomogeneous Beatty sequence*², we permit a constant additive factor inside the floor brackets: $(\lfloor n\alpha + \beta \rfloor)_{n\geq 1}$. The first difference of such a sequence forms a *Sturmian sequence*, which also have been widely studied and have a huge literature [35, 9].

A different generalization of Beatty sequences was proposed by Allouche and Dekking [1]. This concerns sequences of the form $(p \lfloor n\alpha \rfloor + qn + r)_{n \geq 0}$, where p, q, r are natural numbers.

In this paper, we show that if α and β belong to the same quadratic field, then the inhomogeneous Beatty sequence $(\lfloor n\alpha + \beta \rfloor)_{n\geq 1}$ is synchronized, in the sense that there is a finite automaton that takes the Ostrowski representations of n and y in parallel, and accepts if and only if $y = \lfloor n\alpha + \beta \rfloor$. Ostrowski representations are discussed in Section 3; for an introduction and explanation of how automata can use them, see [4]. Synchronized sequences are discussed in Section 4; for a summary, see [43]. As a consequence, we get a new and rather simple proof that first-order logical theory of quadratic Beatty sequences with addition is decidable. History of this result is discussed in Section 2, and our results appear in Section 5. Not only are these first-order statements decidable, but in practice many theorems about individual Beatty sequences can be proved in a matter of seconds with an implementation of the decision procedure in the free software Walnut [36, 45]. Walnut is available at

https://cs.uwaterloo.ca/~shallit/walnut.html .

In Section 7, we apply these results to problems of additive number theory. We show that for each $r \ge 1$ it is decidable whether the set $\{\lfloor n\alpha + \beta \rfloor : n \ge 1\}$ forms an additive basis (or asymptotic additive basis) of order r.

In Section 8 we use these ideas to solve some open problems of Don Reble. In Section 9 we give an explicit description of a sequence studied by Hildebrand, Li, Li, and Xie. In Sections 10 and 11, we solve some open problems of Kimberling. Finally, in Section 12, we prove a number of identities about sequences in the OEIS.

2 Previous work

In this section, we briefly survey previous work. Berstel [7, 8] was apparently the first to show that a finite automaton can recognize the addition relation x + y = z when x, y, and z are given in Fibonacci representation (see Section 3). This was generalized to the case of certain kinds of linear recurrences by Frougny [19, 20, 21]. Mousavi et al. [37] showed how

¹Some authors [14] insist that α must be greater than 1, but we do not. Also, sometimes it is useful to allow the index n to be 0.

²Sometimes called *non-homogeneous Beatty sequence* in the literature; see [17].

to implement Fibonacci representations in the Walnut theorem-proving system and, with it, proved many results in combinatorics on words "purely mechanically" [15, 16].

A related representation, called Pell representation, was studied by Baranwal and Shallit [5]; it is also implemented in Walnut.

More generally, Hieronymi and Terry [26] and, independently Baranwal [3] and Baranwal et al. [4] showed that a finite automaton can do addition in the Ostrowski numeration system (see Section 3) for quadratic numbers. This was implemented in Walnut and also Pecan, another software system [25].

As a consequence, the first-order theory of \mathbb{N} together with the Sturmian sequence $\mathbf{f} = 01001010\cdots$ (the Fibonacci word, corresponding to $(3 - \sqrt{5})/2$) with addition is decidable. This was observed by Bruyère and Hansel [11], building on work of Bruyère et al. [12]. Later, these ideas were used by Mousavi et al. [37] to solve questions about the properties of \mathbf{f} and related words. Also see [24].

More generally, the first-order theory of \mathbb{N} together with the Sturmian sequence for any quadratic number is decidable. This was first observed by Hieronymi and Terry [26], and independently by Baranwal [3] and Baranwal et al. [4].

Similarly, the first-order theory of \mathbb{N} with addition together with the Beatty sequence for the golden ratio is decidable [37, 24, 45]. Recently, Khani and Zarei [31] proved decidability of this theory using a different technique, quantifier elimination.

More generally, the analogous decidability result is true for the Beatty sequence of any quadratic number. This is implied by the results of Hieronymi et al. [25] but not explicitly stated there. For related results, see [30, 22].

The novelty of our results is the simplicity of the method (see, for example, Theorem 4 and Corollary 5) and our applications to solving problems in the theory of Beatty sequences.

3 Ostrowski α -representation

In the famous Fibonacci (or Zeckendorf) system for representing natural numbers [34, 48], every integer is written uniquely as a sum of distinct Fibonacci numbers, subject to the condition that we never use two consecutive Fibonacci numbers. More precisely, if the Fibonacci numbers are (as usual) $F_0 = 0$, $F_1 = 1$, and $F_n = F_{n-1} + F_{n-2}$ for $n \ge 2$, then a natural number n can be expressed as a binary string $e_t \cdots e_1$ with interpretation $n = \sum_{1 \le j \le t} e_i F_{i+1}$. Notice that the most significant digit is written at the left.

The Ostrowski numeration system is a generalization of this idea. Let $\alpha > 0$ be an irrational real number, with continued fraction $[a_0, a_1, a_2, \ldots]$. Define, as usual, the *i*'th convergent p_i/q_i to be $[a_0, a_1, \ldots, a_i]$. Recall (e.g., [23]) that $p_i = a_i p_{i-1} + p_{i-2}$ and $q_i = a_i q_{i-1} + q_{i-2}$.

The Ostrowski α -numeration system is a method for uniquely expressing every natural number n as an integer linear combination of the q_i , namely $\sum_{0 \le i \le t} e_i q_i$. Here, to obtain uniqueness of the representation, the e_i are restricted by the following inequalities:

(a) $0 \le e_i \le a_{i+1}$ for $i \ge 1$;

(b) $e_i = a_{i+1}$ implies $e_{i-1} = 0$ for $i \ge 1$;

(c)
$$0 \le e_0 < a_1$$
.

See [40] and [2, §3.9]. We write the canonical Ostrowski α -representation of n, starting with the most significant digit, and satisfying the rules (a)–(c) above, as $(n)_{\alpha}$. The inverse mapping, which sends a string $x = e_t e_{t-1} \cdots e_0$ to the sum $\sum_{0 \le i \le t} e_i q_i$, is denoted by $[x]_{\alpha}$.

In some cases we need to write the α -representation of pairs or tuples. We do so by working over a product alphabet, and padding the shorter representation at the front with 0's, if needed.

4 Automata, synchronized sequences, and first-order logic

We say that a sequence $(s_n)_{n\geq 0}$ is synchronized if there is a finite automaton that takes the representations of n and y in parallel, and accepts if and only if $y = s_n$. In this paper, the particular representations we use are the Ostrowski α -representations discussed in the previous section.

Synchronized sequences have a number of nice closure properties. These can be found in [43] for automatic sequences, but the same proofs also suffice for sequences defined in terms of Ostrowski representations (provided one knows that the addition relation is synchronized, as proved in [4]).

Proposition 1.

- (a) The sequence $(\lfloor (bn+c)/d \rfloor)$ is synchronized for integers b, c, d with $b \ge \max(1, -c)$ and $d \ge 1$.
- (b) If $(f(n))_{n\geq 1}$ and $(g(n))_{n\geq 1}$ are both synchronized, then so is their termwise sum $(f(n) + g(n))_{n\geq 1}$.
- (c) If $(f(n))_{n\geq 1}$ and $(g(n))_{n\geq 1}$ are both synchronized then so is their composition $f(g(n))_{n\geq 1}$.

From the results of [12] we immediately get the following result.

Theorem 2. There is an algorithm that takes as input a first-order logical formula on natural numbers for f(n), defined in terms of addition, subtraction, multiplication by constants, and synchronized sequences and computes a synchronized automaton for f.

Furthermore, the Walnut theorem-prover can compute these synchronized automata. For more information about Walnut, see [36, 45]. Thus, to prove our main result, we only need to show that the map $n \to |n\alpha + \beta|$ is synchronized.

5 The main result and its implementation

We need a simple lemma.

Lemma 3. Let x be a real number and $c \ge 1$ an integer. Then $\lfloor x/c \rfloor = \lfloor \frac{\lfloor x \rfloor}{c} \rfloor$.

Proof. Using Euclidean division, write x = qc + r for $q \in \mathbb{Z}$ and $0 \le r < c$. Since $0 \le \lfloor r \rfloor < c$, we have

$$\frac{\lfloor x \rfloor}{c} = \frac{qc + \lfloor r \rfloor}{c} = q + \frac{\lfloor r \rfloor}{c}.$$
(1)

Hence $\lfloor \frac{\lfloor x \rfloor}{c} \rfloor = q$. On the other hand x/c = q + r/c, so $\lfloor x/c \rfloor = q$.

Theorem 4. Suppose $0 < \gamma < 1$ is a quadratic irrational with purely periodic continued fraction

 $[0, a_1, a_2, \ldots, a_m, a_1, a_2, \ldots, a_m, a_1, a_2, \ldots, a_m, \ldots].$

Let q_i denote the denominator of the *i*'th convergent to this continued fraction. Then for all integers $n \ge 1$ we have

$$[(n-1)_{\gamma} 0^m]_{\gamma} = q_m(n-1) + q_{m-1} \cdot \lfloor n\gamma \rfloor.$$

Proof. We use the 2×2 matrix approach to continued fractions, pioneered by Hurwitz [28], Frame [18], and Kolden [33], and also discussed (for example) in [2].

This approach uses the fact, easily proved by induction, that

$$\begin{bmatrix} a_0 & 1\\ 1 & 0 \end{bmatrix} \begin{bmatrix} a_1 & 1\\ 1 & 0 \end{bmatrix} \cdots \begin{bmatrix} a_i & 1\\ 1 & 0 \end{bmatrix} = \begin{bmatrix} p_i & p_{i-1}\\ q_i & q_{i-1} \end{bmatrix}$$
(2)

for all $i \ge 0$.

Since $a_i = a_{i+m}$ for all $i \ge 1$, we have

$$\begin{bmatrix} p_{i+m} & p_{i+m-1} \\ q_{i+m} & q_{i+m-1} \end{bmatrix} = \begin{bmatrix} a_0 & 1 \\ 1 & 0 \end{bmatrix} \begin{bmatrix} a_1 & 1 \\ 1 & 0 \end{bmatrix} \cdots \begin{bmatrix} a_{i+m} & 1 \\ 1 & 0 \end{bmatrix}$$
$$= \begin{bmatrix} a_0 & 1 \\ 1 & 0 \end{bmatrix} \begin{bmatrix} a_1 & 1 \\ 1 & 0 \end{bmatrix} \cdots \begin{bmatrix} a_m & 1 \\ 1 & 0 \end{bmatrix} \begin{bmatrix} a_{m+1} & 1 \\ 1 & 0 \end{bmatrix} \cdots \begin{bmatrix} a_{i+m} & 1 \\ 1 & 0 \end{bmatrix}$$
$$= \begin{bmatrix} a_0 & 1 \\ 1 & 0 \end{bmatrix} \begin{bmatrix} a_1 & 1 \\ 1 & 0 \end{bmatrix} \cdots \begin{bmatrix} a_m & 1 \\ 1 & 0 \end{bmatrix} \begin{bmatrix} a_1 & 1 \\ 1 & 0 \end{bmatrix} \cdots \begin{bmatrix} a_i & 1 \\ 1 & 0 \end{bmatrix}$$
$$= \begin{bmatrix} p_m & p_{m-1} \\ q_m & q_{m-1} \end{bmatrix} \begin{bmatrix} q_i & q_{i-1} \\ p_i & p_{i-1} \end{bmatrix},$$

where we have used the fact that $a_0 = 0$.

Let us now compare the entry in row 2 and column 1 of both sides. We get

$$q_{i+m} = q_m q_i + q_{m-1} p_i. (3)$$

Now suppose the Ostrowski γ -representation of n-1 is $\sum_{0 \le i \le t} e_i q_i$. Then from Eq. (3), we get

$$[(n-1)_{\gamma}0^{m}]_{\gamma} = \sum_{0 \le i \le t} e_{i}q_{i+m}$$
$$= \sum_{0 \le i \le t} e_{i}(q_{m}q_{i} + q_{m-1}p_{i})$$
$$= \left(q_{m}\sum_{0 \le i \le t} e_{i}q_{i}\right) + \left(q_{m-1}\sum_{0 \le i \le t} e_{i}p_{i}\right)$$
$$= q_{m}(n-1) + q_{m-1}\lfloor n\gamma \rfloor,$$

as desired. Here, in the last step, we have used [2, Corollary 9.1.14].

Corollary 5. Suppose $0 < \gamma < 1$ is a quadratic irrational with purely periodic continued fraction. Then the sequence $(\lfloor n\gamma \rfloor)_{n\geq 1}$ is γ -Ostrowski synchronized.

Proof. It is easy to see that for each $m \ge 1$ there is a finite automaton shift_m accepting $0^m y$ and $y0^m$ in parallel. We sketch the details. The states are *m*-tuples of integers appearing in the Ostrowski representation of \mathbb{N} , and the initial state and unique final state is 0^m . Here the "meaning" of a state named $f_1 \cdots f_m$ is that these are the last *m* symbols of *y* that the automaton has read. The transition function δ is defined by $\delta(f_1 f_2 \cdots f_m, [f_1, g]) = f_2 \cdots f_m g$.

Since q_m and q_{m-1} are constants, the first-order formula

$$\exists u, v \ n = u + 1 \land \texttt{shift}_m(u, v) \land v = q_{m-1} \cdot z + q_m \cdot u$$

expresses $z = \lfloor n\gamma \rfloor$ as a synchronized function of n, for $n \ge 1$. The result now follows by Theorem 2.

Remark 6. For the special case of $\alpha = (1 + \sqrt{5})/2$, the golden ratio, this was already proved in [37].

We can now prove the two main results of the paper.

Theorem 7. Suppose $0 < \gamma < 1$ is a quadratic irrational with purely periodic continued fraction

 $[0, a_1, a_2, \ldots, a_m, a_1, a_2, \ldots, a_m, a_1, a_2, \ldots, a_m, \ldots],$

and suppose $\alpha, \beta \in \mathbb{Q}(\gamma)$ such that $\alpha \geq 0$ and $\alpha + \beta \geq 0$. Then $(\lfloor n\alpha + \beta \rfloor)_{n \geq 1}$ is synchronized in terms of Ostrowski γ -representations.

Proof. If $\alpha, \beta \in \mathbb{Q}(\gamma)$ satisfy the stated inequalities, then we can write $\alpha = (a + b\gamma)/c$, $\beta = (d + e\gamma)/c$ for some integers a, b, c, d, e with $b \ge 0$ and $c \ge 1$. So

$$\lfloor \alpha n + \beta \rfloor = \lfloor n(a+b\gamma)/c + (d+e\gamma)/c \rfloor$$
$$= \left\lfloor \frac{\lfloor bn\gamma + an + e\gamma + d \rfloor}{c} \right\rfloor$$
$$= \left\lfloor \frac{\lfloor (bn+e)\gamma \rfloor + an + d}{c} \right\rfloor.$$

Now from Corollary 5, we know that $f(n) = \lfloor n\gamma \rfloor$ is γ -Ostrowski synchronized. Hence, by Proposition 1, so is f(bn + e). Hence so is f(bn + e) + an + d. Hence so is $\lfloor \frac{f(bn+e)+an+d}{c} \rfloor$. The result is proved.

Corollary 8. Let γ be a purely periodic quadratic irrational, and let $\alpha, \beta \in \mathbb{Q}(\gamma)$. The first-order theory of the natural numbers with addition and the function $n \to \lfloor n\alpha + \beta \rfloor$ is decidable.

Proof. We combine the result of Theorem 7 with Theorem 2.

Next, we prove a sort of converse to Corollary 5. We need a lemma. Recall that the subword complexity (aka factor complexity) of a word is the function $\rho(n)$ counting the number of distinct length-*n* factors.

Lemma 9. Suppose $0 < \alpha_i, \beta_i < 1$ are real numbers for $1 \le i \le k$. Let

$$\mathbf{s}_i(n) = \lfloor (n+1)\alpha_i + \beta_i \rfloor - \lfloor n\alpha_i + \beta_i \rfloor$$

be the associated Sturmian word with slope α_i and intercept β_i . Form the word **s** over the alphabet $\{0,1\}^k$ by $\mathbf{s}[n] = (\mathbf{s}_1[n], \dots, \mathbf{s}_k[n])$. If the k + 1 numbers

$$1, \alpha_1, \ldots, \alpha_k$$

are linearly independent over \mathbb{Q} , then **s** has subword complexity $(n+1)^k$.

Proof. By the multidimensional version of Kronecker's theorem (see, e.g., [23, Thm. 442]), the linear independence criterion implies that there are integers q and p_i , $1 \le i \le k$ such that $|\sum_{1\le i\le k} q\alpha_i - p_i + b_i| < \epsilon$.

It is known that, for all n, there exists a partition of [0, 1) into n + 1 intervals such that the length-n factor of a Sturmian word starting at position q is completely determined by which interval $q\alpha_i + \beta_i$ lies in, modulo 1 [32]. In other words it is completely determined by which of n + 1 (circular) intervals $q\alpha_i$ lies in. Pick a particular length-n factor in each of the k Sturmian words. For all $i, 1 \le i \le k$, choose a point b_i in the corresponding intervals that is at least ϵ away from the endpoints of the intervals, and choose ϵ sufficiently small. Then by Kronecker's theorem some positive integer q exists that makes $q\alpha_i \mod 1$ at distance at most ϵ from b_i , so it lies in the interior of the specified interval, so the *i*'th Sturmian word beginning at position q is the desired one. Since this is true for each $i, 1 \le i \le k$, all $(n+1)^k$ possibilities for a factor of \mathbf{s} must occur.

We can now prove a sort of converse to Corollary 5.

Theorem 10. Suppose $0 < \gamma < 1$ is a quadratic irrational with purely periodic continued fraction expansion. Suppose the Beatty sequence $(\lfloor n\alpha + \beta \rfloor)_{n \geq 1}$ is γ -Ostrowski synchronized. Then $\alpha \in \mathbb{Q}(\gamma)$. Furthermore, the expression of α in terms of γ is computable. Proof. Suppose $1, \alpha, \gamma$ are linearly independent. If the Beatty sequence $(\lfloor n\alpha + \beta \rfloor)_{n\geq 1}$ is γ -Ostrowski synchronized, then the corresponding Sturmian sequence (possibly recoded over the alphabet $\{0, 1\}$ if necessary) is γ -Ostrowski automatic. This is also true for the Sturmian sequence associated with slope γ and intercept 0. Then the product of these two sequences (over the alphabet $\{0, 1\}^2$) is also γ -Ostrowski automatic. Hence this product has subword complexity $\Theta(n^2)$, but we know that the subword complexity of an automatic sequence is O(n). This contradiction shows that $1, \alpha, \gamma$ are linearly dependent, and hence there are integers a, b, c such that $a\alpha = b\gamma + c$.

It remains to show that we can computably determine a, b, and c. To do so we iterate over all possible integer triples by increasing order of |a| + |b| + |c|. For each triple we construct the an γ -Ostrowski automaton for the Sturmian word with slope $(b\gamma + c)/a$. We then check (using the logical approach) whether the set of factors occurring in Sturmian word is the same as that for the Sturmian word with slope α and intercept β . Since all Sturmian words with the same slope have the same set of factors, this suffices.

6 Beatty sequences in Walnut

We now explain how to use Beatty sequences of quadratic irrationals in Walnut. Suppose α and β lie in a quadratic field $\mathbb{Q}(\gamma)$, where γ is a purely periodic continued fraction.

If the period of γ 's continued fraction starts with 1, rotate the period to get γ' with a period that begins with a number larger than 1. Then γ' generates the same field as γ . This is necessary because Walnut assumes that the continued fraction represents a number in the open interval $(0, \frac{1}{2})$. Rotation of the period only fails when $\gamma = [0, 1, 1, 1, \ldots]$, but in that case we can just use the Fibonacci numeration system instead, based on the real number $[0, 2, 1, 1, 1, \ldots]$.

Next, create the appropriate number system with a command of the form

ost <name> [0] [period]:

This will create a number system called msd_name, where you replace name with any name you like. For example

ost sqrt7 [0] [4 1 1 1]:

Use Theorem 4 to write a Walnut formula for a synchronized automaton accepting n and $\lfloor n\gamma \rfloor$ in parallel. Finally, use the relationship between α, β , and γ , and Theorem 7 to create a Walnut formula for $\lfloor n\alpha + \beta \rfloor$.

As an example, let us obtain the synchronized automaton corresponding to the Beatty sequence where $\alpha = \frac{\sqrt{21}-1}{2}$, $\beta = \frac{\sqrt{21}+3}{4}$. Since $\alpha = [1, 1, 3, 1, 3, ...]$, the periodic portion of the continued fraction is 1, 3, which we then rotate to get [3, 1]. Define $\gamma = [0, 3, 1, 3, 1, 3, 1, ...]$; then $\gamma = (\sqrt{21}-3)/6$. In the corresponding Ostrowski numeration system, the digits can be 0, 1, 2, or 3. We now build the shifting automaton as described in the proof of Corollary 5, and it is depicted in Figure 1. We give it the name shift13 in Walnut.

Next, we obtain the synchronized automaton for $(\lfloor n\gamma \rfloor)_{n\geq 1}$. In Theorem 4, we have $q_m = 4$ and $q_{m-1} = 3$. We use the following Walnut commands:



Figure 1: Shift automaton.

```
ost s13 [0] [3 1]:
# define the numeration system
def beattyg "?msd_s13 (n=0 & z=0) | (Eu,v n=u+1 & $shift13(u,v) & v=3*z+4*u)":
```

This automaton has 32 states. Notice that we have to handle n = 0 as a special case.

We should say something briefly about the syntax of Walnut here. The symbol & represents logical AND, ~ represents logical NOT, | represents logical OR, => is implication, <=> is IFF, E is the existential quantifier, and A is the universal quantifier. The ost command creates an Ostrowski numeration system, the def command defines an automaton for later use, the reg command allows defining a regular expression, and the eval command evaluates a formula with no free variables and reports TRUE or FALSE. We can multiply and do integer division by natural number constants.

Now it is easy to see that $\alpha = 1 + 3\gamma$ and $\beta = (3 + 3\gamma)/2$. Thus in Theorem 7 we take a = 2, b = 6, c = 2, d = 3, and e = 3, and enter the following Walnut command:

```
def beatty "?msd_s13 Eu $beattyg(6*n+3,u) & z=(u+2*n+3)/2":
```

This creates a synchronized automaton for $(\lfloor n\alpha + \beta \rfloor)_{n \ge 1}$, with 59 states.

We now claim that every integer ≥ 12 is the sum of two elements of $(\lfloor n\alpha + \beta \rfloor)_{n\geq 1}$. We can verify this as follows:

eval check2 "?msd_s13 An (n>=12) => Eu,v,n1,n2 u>=1 &
 v>=1 & \$beatty(u,n1) & \$beatty(v,n2) & n=n1+n2":

and Walnut returns TRUE. Note the need to specify that $u, v \ge 1$.

7 The additive theory of Beatty sequences

Given a sequence $(s_i)_{n\geq 1}$ or set $S = \{s_1, s_2, \ldots\}$ of natural numbers, a fundamental question of additive number theory is to decide which integers have representations as r-fold sums of the s_i [38, 39]. More precisely, we define the set of r-fold sums $A_s(r)$ as follows:

 $A_s(r) = \{ n \in \mathbb{N} : \exists i_1, i_2, \dots, i_r \text{ such that } n = s_{i_1} + s_{i_2} + \dots + s_{i_r} \}.$

We say that $(s_i)_{n\geq 1}$ forms an additive basis of order r if $A_s(r) = \mathbb{N}$. We say that $(s_i)_{n\geq 1}$ forms an asymptotic additive basis of order r if $\mathbb{N} \setminus A_s(r)$ is finite.

In this section we generalize some recent results on the additive theory of Beatty sequences [29, 41, 44, 13].

Theorem 11. Let γ be a real quadratic irrational, and suppose $\alpha, \beta \in \mathbb{Q}(\gamma)$. Then the inhomogeneous Beatty sequence $(\lfloor \alpha n + \beta \rfloor)_{n \geq 0}$ forms an asymptotic additive basis of order h for some integer $h \geq 1$. Furthermore, the smallest such h is effectively computable, and one can also decide if $(\lfloor \alpha n + \beta \rfloor)_{n \geq 0}$ forms an additive basis.

Proof. By a classical theorem (see Nathanson [39, p. 366, Theorems 11.6 and 11.7]) every Beatty sequence forms an asymptotic additive basis where an upper bound on the number of terms needed to represent every sufficiently large number is (effectively) computable. Now we just write first-order statements that every sufficiently large integer is a sum of h terms of the Beatty sequence, and evaluate them using the algorithm of Theorem 2, until we find the smallest h. The algorithm is guaranteed to terminate by Nathanson's theorem.

Similarly we can write a first-order statement that every natural number is a sum of some number of terms. $\hfill \Box$

Remark 12. The same result applies to the generalized Beatty sequences of Allouche and Dekking [1].

We now illustrate the ideas for a particular example. We use the free software Walnut, which can prove theorems about Ostrowski-automatic sequences, if they are stated in first-order logic. For more details about Walnut, see [36, 45].

Let $\varphi = (1 + \sqrt{5})/2$, the golden ratio, and consider the sequence $\mathbf{c} = (n\varphi + \frac{1}{2})_{n\geq 0}$. This is sequence <u>A007067</u> in the OEIS [46], and was originally studied by Johann Bernoulli [47, pp. 66–69].

Theorem 13. The sequence \mathbf{c} is φ -synchronized.

Proof. Starting from the known automaton (called **phin**) for the Beatty sequence $(\lfloor n\varphi \rfloor)_{n\geq 0}$ (see, e.g., [45, §10.11]), we can get an automaton for **c** as follows (in Walnut)

reg shift {0,1} {0,1} "([0,0] | [0,1] [1,1]*[1,0])*": def phin "?msd_fib (s=0&n=0) | Ex \$shift(n-1,x) & s=x+1": def eta "?msd_fib Er \$phin(2*n,r) & z=(r+1)/2":

This produces the synchronized automaton for \mathbf{c} in Figure 2.



Figure 2: Synchronized automaton for $(\lfloor n\varphi + \frac{1}{2} \rfloor)_{n \ge 0}$.

Theorem 14. Every integer $n \neq 1$ can be written as the sum of two elements of **c**.

Proof. We use the following Walnut commands:

```
def iseta "?msd_fib En $eta(n,s)":
eval test "?msd_fib An (n!=1) <=> Ei, j $iseta(i) & $iseta(j) & n=i+j":
and this returns TRUE. The result is proved.
```

8 Proofs of some conjectures of Reble

As further applications we prove some conjectures of Don Reble regarding the OEIS sequences <u>A189377</u>, <u>A189378</u>, and <u>A189379</u>.

These conjectures relate to the existence of monochromatic arithmetic progressions of length 3 in the Fibonacci word **f**, defined by the fixed point of the morphism $0 \to 01$, $1 \to 0$, or, alternatively, as the Sturmian word corresponding to $\alpha = (3 - \sqrt{5})/2$. More specifically, we are interested in those r for which

- (i) there exists *i* such that $\mathbf{f}[i] = \mathbf{f}[i+r] = \mathbf{f}[i+2r] = 0$
- (ii) there exists j such that $\mathbf{f}[j] = \mathbf{f}[j+r] = \mathbf{f}[j+2r] = 1$.

Reble's conjectures can be stated as follows.

Theorem 15. Let r = 2, $s = (\sqrt{5} - 1)/2$, and $t = (1 + \sqrt{5})/2$. Define

$$a(n) = n + \lfloor ns/r \rfloor + \lfloor nt/r \rfloor$$

$$b(n) = n + \lfloor nr/s \rfloor + \lfloor nt/s \rfloor$$

$$c(n) = n + \lfloor nr/t \rfloor + \lfloor ns/t \rfloor.$$

Then

- (a) there exists m such that m = a(n) iff (i) holds for r = n + 1 but (ii) does not for r = n + 1;
- (b) there exists m such that m = b(n) iff neither (i) nor (ii) holds for r = n + 1;
- (c) there exists m such that m = c(n) iff both (i) and (ii) hold for r = n + 1.

Proof. We implement the formulas above in Walnut. Here F is Walnut's abbreviation for the Fibonacci word, indexed starting at position 0.

```
def three0 "?msd_fib Ei (F[i]=@0) & (F[i+n]=@0) & (F[i+2*n]=@0)":
def three1 "?msd_fib Ei (F[i]=@1) & (F[i+n]=@1) & (F[i+2*n]=@1)":
def fibsr "?msd_fib Er $phin(n,r) & s=(r-n)/2":
def fibtr "?msd_fib Er $phin(n,r) & s=r/2":
def fibrs "?msd_fib $phin(2*n,s)":
def fibts "?msd_fib Er $phin(n,r) & s=r+n":
def fibrt "?msd_fib Er $phin(2*n,r) & s=r-2*n":
def fibst "?msd_fib Er $phin(n,r) & s=2*n-(r+1)":
def a189377 "?msd_fib En,x,y $fibsr(n,x) & $fibtr(n,y) & z=n+x+y":
def a189378 "?msd_fib En,x,y $fibrs(n,x) & $fibts(n,y) & z=n+x+y":
def a189379 "?msd_fib En,x,y $fibrt(n,x) & $fibst(n,y) & z=n+x+y":
def reble1 "?msd_fib $three0(n) & ~$three1(n)":
def reble2 "?msd_fib ~$three0(n) & ~$three1(n)":
def reble3 "?msd_fib $three0(n) & $three1(n)":
eval rebleconj1 "?msd_fib An (n>=1) => ($reble1(n+1) <=> $a189377(n))":
eval rebleconj2 "?msd_fib An (n>=1) => ($reble2(n+1) <=> $a189378(n))":
eval rebleconj3 "?msd_fib An (n>=1) => ($reble3(n+1) <=> $a189379(n))":
```

When we run them, we get TRUE as the result. The entire computation takes less than a second. $\hfill \Box$

9 A sequence of Hildebrand, Li, Li, and Xie

Hildebrand, Li, Li, and Xie [27] studied a certain sequence $\tilde{c}(n)$, defined as follows: let $a(n) = \lfloor n\varphi^4 \rfloor$, $b(n) = \lfloor n\varphi^3 \rfloor$, $c(n) = \lfloor n\varphi \rfloor$, and $\tilde{c}(n)$ be the *n*'th positive integer that is not contained in the union $\{a(i) : i \ge 1\} \cup \{b(i) : i \ge 1\}$. They showed that $0 \le c(n) - \tilde{c}(n) \le 2$. We can reprove this and do a little more, using our techniques:

Theorem 16. The Fibonacci automaton displayed in Figure 3, on input n in the Fibonacci numeration system, computes $c(n) - \tilde{c}(n)$.



Figure 3: Fibonacci automaton for $c(n) - \tilde{c}(n)$. A state labeled q/a has output a.

Proof. We use the fact that $\varphi^4 = 3\varphi + 2$ and $\varphi^3 = 2\varphi + 1$. We can therefore define synchronized automata for a(n), b(n), and c(n), as follows:

```
def a "?msd_fib Em $phin(3*n,m) & z=m+2*n":
def b "?msd_fib Em $phin(2*n,m) & z=m+n":
def c "?msd_fib $phin(n,z)":
```

Now $\tilde{c}(n)$ is defined as the *n*'th positive integer not contained in the union $\{a(i) : i \ge 1\} \cup \{b(i) : i \ge 1\}$. First, let us check that there is no intersection between the images of *a* and *b*:

```
eval no_inter "?msd_fib ~Em,n,x m>1 & n>1 & $a(m,x) & $b(n,x)":
```

Next, let's define inverses to a(n) and b(n) as follows: we choose $a^{-1}(m)$ such that it is the least value for which $a(a^{-1}(m)) \leq m < a(a^{-1}(m) + 1)$, and the analogous inequality for $b^{-1}(n)$:

```
def ai "?msd_fib Et,u $a(z,t) & $a(z+1,u) & t<=m & m<u":
# ai(m) = z, t=a(ai(m)), u=a(ai(m)+1)
def ainv "?msd_fib $ai(m,z) & Ax (x<z & x>0) => ~$ai(m,x)":
# the least m with the property
```

def bi "?msd_fib Et,u \$b(z,t) & \$b(z+1,u) & t<=m & m<u":
bi(m) = z, t=b(bi(m)), u=b(bi(m)+1)
def binv "?msd_fib \$bi(m,z) & Ax (x<z & x>0) => ~\$bi(m,x)":
the least m with the property

Next, we can define $\tilde{c}(n)$ as the least y such that $a^{-1}(y) + b^{-1}(y) + n = y$:

```
def ctw "?msd_fib Es,t $ainv(y,s) & $binv(y,t) & s+t+n=y":
def ctwid "?msd_fib $ctw(n,y) & Az (z<y) => ~$ctw(n,z)":
```

Next we can check that $\tilde{c}(n)$ is indeed the complementary sequence.

```
eval check4 "?msd_fib Ax (x>0) => (En (n>=1) & ($a(n,x)|$b(n,x))) <=>
  (~En (n>=1) & $ctwid(n,x))":
```

ctwid is the complementary sequence of a(n) union b(n)

Next, we check that the inequality $0 \le c(n) - \tilde{c}(n) \le 2$ holds:

```
eval check5 "?msd_fib An,x,y ($phin(n,x) & $ctwid(n,y)) => (x=y|x=y+1|x=y+2)":
```

and Walnut returns TRUE.

Finally, we combine the individual automata for the three possible values of $c(n) - \tilde{c}(n)$ to get a single automaton that computes the difference:

```
def has11 {0,1} "(0+1)*11(0+1)*":
def diff0 "?msd_fib Ex $phin(n,x) & $ctwid(n,x)":
def diff1 "?msd_fib Ex $phin(n,x+1) & $ctwid(n,x)":
def diff2 "?msd_fib Ex $phin(n,x+2) & $ctwid(n,x)":
```

```
combine diff has11=-1 diff1=1 diff2=2:
```

which gives us the automaton in Figure 3, once the dead state corresponding to invalid Fibonacci representations is removed. $\hfill \Box$

10 The Kimberling swappage problem

Let $L(n) = \lfloor n\varphi \rfloor$ be the lower Wythoff sequence and $U(n) = \lfloor n\varphi^2 \rfloor$ be the upper Wythoff sequence. Kimberling defined the "lower even swappage sequence" $V_{\rm le}(n)$ as follows: if U(n)is even, then $V_{\rm le}(n) := U(n)$. Otherwise, if U(n) is odd, then $V_{\rm le}(n) := L(i)$, where *i* is the least index such that U(n-1) < L(i) < U(n+1) and L(i-1) < U(n) < L(i+1). Here $(V_{\rm le}(n))$ is OEIS sequence A141104. He conjectured that the sequence $S(n) := V_{\rm le}(n)/2$ consists of those positive integers not in the Beatty sequence $(\lfloor n\varphi^3 \rfloor)_{n\geq 1}$, which was later proved by Russo and Schwiebert [42].

We can use Walnut to provide an alternate proof of Kimberling's conjecture. The ideas are those similar to those of the previous section. First, using the ideas in [45, §10.15], we guess a synchronized automaton leswap for $V_{\rm le}(n)$ (it has 34 states), and we then verify it satisfies the definition.

```
def chk1 "?msd_fib An Ex $leswap(n,x)":
def chk2 "?msd_fib An ~Ex1,x2 x1!=x2 & $leswap(n,x1) & $swap(n,x2)":
def odd "?msd_fib Ex n=2*x+1":
def even "?msd_fib Ex n=2*x":
def index "?msd_fib Aa,b,c,d,e,f ($phin(i-1,a) & $phin(i,b) & $phin(i+1,c) &
    $phi2n(n-1,d) & $phi2n(n,e) & $phi2n(n+1,f)) => (a<e & e<c & d<b & b<f)":
def leastindex "?msd_fib $index(i,n) & Aj (1<=j & j<i) => ~$index(j,n)":
eval checkeven "?msd_fib An,x,y ($phi2n(n,x) & $even(x) & $leswap(n,y))
    => x=y":
eval checkodd "?msd_fib Ai,n,x,y,z (i>=1 & n>=1 & $phi2n(n,x) & $odd(x) &
    $leswap(n,y) & $leastindex(i,n) & $phin(i,z)) => y=z":
    Next, we verify the characterization in terms of the Beatty sequence:
```

```
eval kimber "?msd_fib Ax (x>0) => (En (n>=1) & $phi3n(n,x))
<=> (~En,y (n>=1) & $leswap(n,y) & 2*x=y)":
```

and Walnut returns TRUE.

The same techniques can be used to prove the conjectures stated in the OEIS about analogous sequences, namely

- the "upper even" sequence $V_{ue}(n) \underline{A141105}$;
- the "lower odd" sequence $V_{\rm lo}(n)$ <u>A141106</u>; and
- the "upper odd" sequence $V_{uo}(n)$ <u>A141107</u>.

In particular, we can prove the following theorem with Walnut:

Theorem 17. We have

- (a) $V_{ue}(n) = 2\lfloor \frac{\varphi+1}{2}n + \frac{1}{2} \rfloor;$
- (b) $V_{\text{le}}(n) = 2\lfloor \frac{\varphi+1}{2}n \rfloor;$
- (c) $V_{\rm uo}(n) = 2\lfloor \frac{\varphi+1}{2}n \rfloor + 1;$
- (d) $V_{\text{lo}}(n) = 2\lfloor \frac{\varphi+1}{2}n + \frac{1}{2} \rfloor 1.$

We omit the details.

11 Sums-complement sets

Clark Kimberling has studied (see <u>A276871</u> in the OEIS) what he calls "sums-complement" sets for irrational numbers α . These are the positive integers that cannot be written as the difference of two elements of the Beatty sequence $|n\alpha|$. More precisely,

 $S_{\alpha} := \{ n \ge 1 : \neg \exists i, j \ n = \lfloor i\alpha \rfloor - \lfloor j\alpha \rfloor \}.$

Recall that a set S is said to be γ -Ostrowski automatic if there is a finite automaton accepting the γ -Ostrowski representation of n as input and accepting if and only if $n \in S$. Then we have the following proposition.

Theorem 18. Suppose $0 < \gamma < 1$ is a quadratic irrational with purely periodic continued fraction

 $[0, a_1, a_2, \ldots, a_m, a_1, a_2, \ldots, a_m, a_1, a_2, \ldots, a_m, \ldots],$

and suppose $\alpha \in \mathbb{Q}(\gamma)$ such that $\alpha \geq 0$. Then the sums-complement set S_{α} is γ -Ostrowski automatic.

Proof. The claim that $n \in S_{\alpha}$ is a first-order statement in terms of the Beatty sequence for α .

As a consequence, we may use Walnut to obtain explicit Ostrowski automata for the quadratic numbers tabulated in <u>A276871</u>.

Example 19. Let $\gamma = [0, \overline{4, 1, 1, 1}] = (\sqrt{7} - 2)/3$ and $\alpha = \sqrt{7} = 3\gamma + 2$. Then there is an γ -Ostrowski automaton of 6961 states that recognizes the sums-complement set S_{α} , which is sequence <u>A276873</u> in the OEIS. This automaton can be computed with the following Walnut code:

12 Sequences related to $\sqrt{2}$

In this section we find Ostrowski automata for seven sequences from the OEIS related to $\sqrt{2}$, and we re-prove a number of conjectures related to these sequences. The first few terms are given below in Table 1.

n	0	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16
<u>A097508</u> (n)	0	0	0	1	1	2	2	2	3	3	4	4	4	5	5	6	6
A001951(n)	0	1	2	4	5	7	8	9	11	12	14	15	16	18	19	21	22
A003151(n)	0	2	4	7	9	12	14	16	19	21	24	26	28	31	33	36	38
A276862(n)	2	2	3	2	3	2	2	3	2	3	2	2	3	2	3	2	3
A097509(n)	3	2	3	2	3	2	2	3	2	3	2	2	3	2	3	2	3
A080754(n)	0	3	5	8	10	13	15	17	20	22	25	27	29	32	34	37	39
A082844(n)	2	3	2	3	2	2	3	2	3	2	2	3	2	3	2	3	2

Table 1: Seven OEIS sequences related to $\sqrt{2}$.

We start with <u>A097508</u>(n), whose definition is $\lfloor n(\sqrt{2} - 1) \rfloor$. In the corresponding Ostrowski numeration system, the digits can be 0, 1 or 2. The shifting automaton **shift1** can be built using Corollary 5, and it is depicted in Figure 4.



Figure 4: Shift automaton.

Using Theorem 5, we can obtain the synchronized automaton for $\lfloor n(\sqrt{2}-1) \rfloor$ from the following Walnut commands.

ost s2 [0] [2]:
define the numeration system
def a097508 "?msd_s2 (n=0 & z=0) | (Eu,v n=u+1 & \$shift1(u,v) & v=z+2*u)":

Next we turn to OEIS sequence $\underline{A001951}(n)$, defined by $\lfloor n\sqrt{2} \rfloor$. We apply Theorem 7 using a = 1, b = 1, c = 1, d = 0, and e = 0 to create the automaton for $\underline{A001951}$ using the following Walnut command.

def a001951 "?msd_s2 Eu \$a097508(n,u) & z=u+n":

Next we turn to OEIS sequence $\underline{A003151}(n)$, defined by $\lfloor n(\sqrt{2}+1) \rfloor$. We apply Theorem 7 using a = 2, b = 1, c = 1, d = 0, and e = 0 to create the automaton for $\underline{A003151}$ using the following Walnut command.

def a003151 "?msd_s2 Eu \$a097508(n,u) & z=u+2*n":

Next we look at OEIS sequence $\underline{A276862}(n)$, defined to be the first difference of $\underline{A003151}$. We can create a synchronized automaton for it as follows:

def a276862 "?msd_s2 Eu,v \$a003151(n,u) & \$a003151(n+1,v) & z+u=v":

Next we look at OEIS sequence <u>A097509</u>, defined to be the number of times that n occurs as a value of <u>A097508</u>. It is clear that this number is either 2 or 3, so we can build the synchronized automaton as follows:


```
def a097509 "?msd_s2 (z=3 & $three_times(n)) | (z=2 & ~$three_times(n))":
```

Michel Dekking conjectured in 2020 that $\underline{A097509}(n) = \underline{A276862}(n)$ for $n \ge 1$ (see the first comment in $\underline{A276862}$). We can prove this with the following Walnut code:

eval dek "?msd_s2 An,x (n>=1) => (\$a097509(n,x) <=> \$a276862(n,x))":

Next we turn to OEIS sequence <u>A080754</u>(n), which is defined to be $[n(\sqrt{2}+1)]$.

def a080754 "?msd_s2 (n=0 & z=0) | (n>0 & Eu \$a003151(n,u) & z=u+1)":

We can now find a relationship between A097509 and A080754; namely,

$$\underline{A097509}(i) = \underline{A080754}(i+1) - \underline{A080754}(i) \tag{4}$$

for all $i \geq 0$.

Finally, we consider $a(n) := \underline{A082844}(n)$. It has a rather complicated definition, as follows:

$$a(t) = \begin{cases} 3, & \text{if } t = 1; \\ 2, & \text{if } t = 0, 2; \\ a(n), & \text{if } t = a(1) + \dots + a(n); \\ 2, & \text{if } a(t-1) = 3 \text{ and } \neg \exists n \ge 0 \quad t = a(1) + \dots + a(n); \\ 3, & \text{if } a(t-1) = 2 \text{ and } \neg \exists n \ge 0 \quad t = a(1) + \dots + a(n). \end{cases}$$
(5)

To get a formula for a(n), we use a somewhat roundabout approach. Namely, we define the sequence b(n) as <u>A097509</u>(n + 1) for $n \ge 0$, and then show b(n) satisfies the recurrence Eq. (5). This will show that in fact a(n) = b(n) for all n.

This approach requires computing $b(1) + \cdots + b(n)$, which we do as follows:

$$\sum_{1 \le i \le n} b(i) = \sum_{1 \le i \le n} \underline{A097509}(i+1)$$

=
$$\sum_{1 \le i \le n} (\underline{A080754}(i+2) - \underline{A080754}(i+1))$$

=
$$\underline{A080754}(n+2) - \underline{A080754}(2)$$

=
$$\underline{A080754}(n+2) - 5.$$

We then use the following Walnut code:

```
def b "?msd_s2 $a097509(n+1,z)":
eval check1 "?msd_s2 An,u,v,w ($a080754(n+2,u) & $b(u-5,v) & $b(n,w))
    => v=w":
eval check2 "?msd_s2 At ($b(t-1,3) & ~En $a080754(n+2,t+5)) => $b(t,2)":
eval check3 "?msd_s2 At ($b(t-1,2) & ~En $a080754(n+2,t+5)) => $b(t,3)":
```

Walnut returns TRUE for all cases, which proves the equality of a(n) and b(n).

This also proves the conjecture of Benedict Irwin from 2016, to the effect that

A082844(n) = A097509(n+1)

for all $n \ge 0$ (see the comments to <u>A082844</u>). For more information, see [10, 2nd solution to problem B6].

Acknowledgments

We are grateful to Aaron Barnoff for some corrections and for a question that led to the formulation and proof of Theorem 5, and to Jean-Paul Allouche for suggesting we look at the papers [27, 42, 22].

References

- J.-P. Allouche and F. M. Dekking. Generalized Beatty sequences and complementary triples. Moscow J. Combinatorics and Number Theory 8 (2019), 325–341.
- [2] J.-P. Allouche and J. Shallit. Automatic Sequences: Theory, Applications, Generalizations. Cambridge University Press, 2003.

- [3] A. R. Baranwal. Decision algorithms for Ostrowski-automatic sequences. Master's thesis, University of Waterloo, School of Computer Science, 2020.
- [4] A. R. Baranwal, L. Schaeffer, and J. Shallit. Ostrowski-automatic sequences: theory and applications. *Theoret. Comput. Sci.* 858 (2021), 122–142.
- [5] A. R. Baranwal and J. Shallit. Critical exponent of infinite balanced words via the Pell number system. In R. Mercaş and D. Reidenbach, editors, WORDS 2019, Vol. 11682 of Lecture Notes in Computer Science, pp. 80–92. Springer-Verlag, 2019.
- [6] S. Beatty. Problem 3173. Amer. Math. Monthly 33 (1926), 159. Solution, 34 (1927), 159–160.
- [7] J. Berstel. Fonctions rationnelles et addition. In M. Blab, editor, Théorie des Langages, École de printemps d'informatique théorique, pp. 177–183. LITP, 1982.
- [8] J. Berstel. Fibonacci words—a survey. In G. Rozenberg and A. Salomaa, editors, *The Book of L*, pp. 13–27. Springer-Verlag, 1986.
- [9] J. Berstel, A. Lauve, C. Reutenauer, and F. V. Saliola. Combinatorics on Words: Christoffel Words and Repetitions in Words, Vol. 27 of CRM Monograph Series. Amer. Math. Soc., 2009.
- [10] M. Bhargava, K. Kedlaya, and L. Ng. Solutions to the 81st William Lowell Putnam mathematical competition. Available at https://kskedlaya.org/putnam-archive/ 2020s.pdf, 2021.
- [11] V. Bruyère and G. Hansel. Bertrand numeration systems and recognizability. *Theoret. Comput. Sci.* 181 (1997), 17–43.
- [12] V. Bruyère, G. Hansel, C. Michaux, and R. Villemaire. Logic and p-recognizable sets of integers. Bull. Belgian Math. Soc. 1 (1994), 191–238. Corrigendum, Bull. Belg. Math. Soc. 1 (1994), 577.
- [13] F. M. Dekking. Sumsets and fixed points of substitutions. Proc. Amer. Math. Soc., Ser. B 9 (2022), 393–403.
- [14] F. M. Dekking. On Hofstadter's G-sequence. J. Integer Sequences 26 (2023), 23.9.2 (electronic). https://cs.uwaterloo.ca/journals/JIS/VOL26/Dekking2/dek9.html
- [15] C. F. Du, H. Mousavi, E. Rowland, L. Schaeffer, and J. Shallit. Decision algorithms for Fibonacci-automatic words, II: Related sequences and avoidability. *Theoret. Comput. Sci.* 657 (2017), 146–162.
- [16] C. F. Du, H. Mousavi, L. Schaeffer, and J. Shallit. Decision algorithms for Fibonacciautomatic words III: Enumeration and abelian properties. *Internat. J. Found. Comp. Sci.* 27 (2016), 943–963.

- [17] A. S. Fraenkel. The bracket function and complementary sets of integers. Canad. J. Math. 21 (1969), 6–27.
- [18] J. S. Frame. Continued fractions and matrices. Amer. Math. Monthly 56 (1949), 98–103.
- [19] C. Frougny. Linear numeration systems of order two. Inform. Comput. 77 (1988), 233-259.
- [20] C. Frougny. Fibonacci representations and finite automata. *IEEE Trans. Inform. Theory* 37 (1991), 393–399.
- [21] C. Frougny. Representations of numbers and finite automata. Math. Systems Theory 25 (1992), 37–60.
- [22] A. Günaydın and M. Ozsahakyan. Expansions of the group of integers by Beatty sequences. Ann. Pure Appl. Logic 173 (2022), 103062.
- [23] G. H. Hardy and E. M. Wright. An Introduction to the Theory of Numbers. Oxford Univ. Press, 5th edition, 1985.
- [24] P. Hieronymi. Expansions of the ordered additive group of real numbers by two discrete subgroups. J. Symbolic Logic 81 (2016), 1007–1027.
- [25] P. Hieronymi, D. Ma, R. Oei, L. Schaeffer, C. Schulz, and J. Shallit. Decidability for Sturmian words. In F. Manea and A. Simpson, editors, 30th EACSL Annual Conference on Computer Science Logic (CSL 2022), Leibniz International Proceedings in Informatics, pp. 24:1–24:23. Schloss Dagstuhl—Leibniz-Zentrum für Informatik, 2022.
- [26] P. Hieronymi and A. Terry, Jr. Ostrowski numeration systems, addition, and finite automata. Notre Dame J. Formal Logic 59 (2018), 215–232.
- [27] A. J. Hildebrand, Х. Li, J. Li, and Υ. Xie. Almost Beatty par- $\mathbf{22}$ (electronic). titions. J. Integer Sequences (2019),Article 19.4.6https://cs.uwaterloo.ca/journals/JIS/VOL22/Hildebrand/hilde2.html
- [28] A. Hurwitz and N. Kritikos. *Lectures on Number Theory*. Springer-Verlag, 1986.
- [29] S. Kawsumarng, T. Khemaratchatakumthorn, P. Noppakaew, and P. Pongsriiam. Sumsets associated with Wythoff sequences and Fibonacci numbers. *Period. Math. Hung.* 82 (2021), 98–113.
- [30] M. Khani, A. N. Valizadeh, and A. Zarei. Model-completeness and decidability of the additive structure of integers expanded with a function for a Beatty sequence. Arxiv preprint arXiv:2110.01673v3 [math.LO]. Available at https://arxiv.org/abs/2110. 01673v3, 2022.
- [31] M. Khani and A. Zarei. The additive structure of integers with the lower Wythoff sequence. Archive for Mathematical Logic 62 (2023), 225–237.

- [32] D. E. Knuth. Solution to problem E 2307. Amer. Math. Monthly 79 (1972), 773–774.
- [33] K. Kolden. Continued fractions and linear substitutions. Archiv for Mathematik og Naturvidenskab 50 (1949), 141–196.
- [34] C. G. Lekkerkerker. Voorstelling van natuurlijke getallen door een som van getallen van Fibonacci. Simon Stevin 29 (1952), 190–195.
- [35] M. Lothaire. Algebraic Combinatorics on Words, Vol. 90 of Encyclopedia of Mathematics and its Applications. Cambridge University Press, 2002.
- [36] H. Mousavi. Automatic theorem proving in Walnut. Arxiv preprint arXiv:1603.06017 [cs.FL], available at http://arxiv.org/abs/1603.06017, 2016.
- [37] H. Mousavi, L. Schaeffer, and J. Shallit. Decision algorithms for Fibonacci-automatic words, I: Basic results. *RAIRO Inform. Théor. App.* **50** (2016), 39–66.
- [38] M. B. Nathanson. Additive Number Theory—The Classical Bases. Springer-Verlag, 1996.
- [39] M. B. Nathanson. *Elementary Methods in Number Theory*. Springer, 2000.
- [40] A. Ostrowski. Bemerkungen zur Theorie der Diophantischen Approximationen. Abh. Math. Sem. Hamburg 1 (1922), 77–98,250–251. Reprinted in Collected Mathematical Papers, Vol. 3, pp. 57–80.
- [41] P. N. Phunphayap, P. Pongsriiam, and J. Shallit. Sumsets associated with Beatty sequences. *Discrete Math.* 345 (2022), 112810.
- [42] V. Russo and L. Schwiebert. Beatty sequences, Fibonacci numbers, and the golden ratio. *Fibonacci Quart.* 49 (2011), 151–154.
- [43] J. Shallit. Synchronized sequences. In T. Lecroq and S. Puzynina, editors, WORDS 2021, Vol. 12847 of Lecture Notes in Computer Science, pp. 1–19. Springer-Verlag, 2021.
- [44] J. Shallit. Sumsets of Wythoff sequences, Fibonacci representation, and beyond. Period. Math. Hung. 84 (2022), 37–46.
- [45] J. Shallit. The Logical Approach To Automatic Sequences: Exploring Combinatorics on Words with Walnut, Vol. 482 of London Math. Society Lecture Note Series. Cambridge University Press, 2023.
- [46] N. J. A. Sloane et al. The On-Line Encyclopedia of Integer Sequences. Electronic resource, available at https://oeis.org, 2024.
- [47] B. A. Venkov. *Elementary Number Theory*. Wolters-Noordhoff, 1970.
- [48] E. Zeckendorf. Représentation des nombres naturels par une somme de nombres de Fibonacci ou de nombres de Lucas. Bull. Soc. Roy. Liège 41 (1972), 179–182.